<commit_before>{pkgs, ...}:

let
  fragile = pkgs.callPackage ./. {};
in {
  environment.systemPackages = [ fragile ];
  security.sudo.extraConfig = ''
    Cmnd_Alias FRAGILE = ${fragile}/bin/fragile
    Defaults!FRAGILE env_keep += NIX_PATH
    %nixbld ALL = NOPASSWD: FRAGILE
  '';
}<commit_msg>FIX: Proper example module with good suPath<commit_after>{config, lib, pkgs, ...}:

with lib;

let
  cfg = config.services.fragile;
  fragile = pkgs.callPackage ./. { suPath = "${config.security.wrapperDir}/su"; };
in {
  options = {
    services.fragile.enable = mkOption {
      type = types.bool;
      default = false;
      description = ''
        Whether to enable fragile, placing it in the system packages.
        This may have security implications.
      '';
    };

    services.fragile.permitNixbldSudo = mkOption {
      type = types.bool;
      default = true;
      description = ''
        Whether to permit passwordless sudo access to fragile for users in the
        nixbld group.
      '';
    };
  };

  config = mkIf cfg.enable {
    environment.systemPackages = [ fragile ];

    security.sudo.extraConfig = mkIf cfg.permitNixbldSudo ''

      # services.fragile.permitNixbldSudo
      Cmnd_Alias FRAGILE = ${fragile}/bin/fragile
      Defaults!FRAGILE env_keep += NIX_PATH
      %nixbld ALL = NOPASSWD: FRAGILE
    '';
  };
}
{'fraction_non_alphanumeric': 0.11804008908685969, 'fraction_numerical': 0.0, 'mean_word_length': 2.6630434782608696, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1393216', 'n_tokens_mistral': 453, 'n_tokens_neox': 436, 'n_words': 100}